**frames modify** — Modify a set of frames on disk



## Description

`frames modify` modifies a frameset (`.dtas`) file by adding frames, dropping frames, or replacing the content of existing frames in the file.

## Quick start

Add frames `A`, `B`, and `C` to file `myframeset.dtas`

```
frames modify using myframeset, add(A B C)
```

Drop frames `A` and `B` from file `myframeset.dtas`

```
frames modify using myframeset, drop(A B)
```

## Syntax

*Add frames to a frameset on disk*

`frames modify using` *filename*`, add(`*framelist* [ `, replace` ]`) [`*options*`]`

*Drop frames from a frameset on disk*

`frames modify using` *filename*`, drop(`*framelist*`)`

If *filename* is specified without an extension, `.dtas` is assumed. If *filename* contains embedded spaces or other special characters, enclose it in double quotes.

| *options* | Description |
|---|---|
| `nolabel` | omit value labels from the added frames |
| `orphans` | save value labels in added frames, even if they are not attached to a variable |
| `emptyok` | add specified frames even if they have zero observations and zero variables |

## Options

add(*framelist*[ , replace]) specifies the frames in memory to be added to the frameset. *framelist* is a list of frame names separated by a space. To add all frames in memory to the frameset, specify add(_all). Either add() or drop(), but not both, must be specified.

replace permits frames modify to overwrite frames that already exist in the frameset.

drop(*framelist*) specifies the frames to be dropped from the frameset. *framelist* is a list of frame names separated by a space. Either drop() or add(), but not both, must be specified.

nolabel specifies that value labels be omitted when adding frames to the frameset.

orphans specifies that all value labels be saved with the frames to be added, including those not attached to any variable.

emptyok specifies that frames be added to the frameset even if they contain zero observations and zero variables.

## Remarks and examples

frames modify allows you to conveniently modify a frameset (.dtas) file. You can add a list of frames to the frameset, or drop a list of frames from the frameset, without loading the entire frameset into memory.

frames modify is useful when you have already saved a set of frames with frames save but wish to modify its contents. However, frames modify has the potential to break linkages, if they exist, between frames in the frameset. Adding a frame that does not previously exist in the frameset will not affect any existing links in the frameset. However, if a frame in the frameset was saved with links to other frames, the linkages will be dropped if you replace the frame without reestablishing the link. If you are not sure about existing linkages, you should load the frameset with frames use and examine linkages before using frames modify.

### ▷ Example 1: Modify an existing frame in the frameset

In frames save, we saved frames census and housing in myframeset.dtas. Below, we re-create that file:

```
. frame create census
. frame change census
. sysuse census
(1980 Census data by state)
. frame create housing
. frame change housing
. webuse hsng
(1980 Census housing data)
. frlink 1:1 state, frame(census)
(all observations in frame housing matched)
. frames save myframeset, frames(housing) linked
file myframeset.dtas saved
```

Suppose that we wish to modify the contents of the housing frame. This is the current frame. Below, we drop two variables that we are not interested in, and then we replace the contents of the housing frame in myframeset.dtas.

```
. drop popden popgrow
. frames modify using myframeset, add(housing, replace)
frame housing replaced
file myframeset.dtas saved
```

frames modify reports that the housing frame was replaced and that the frameset file myframeset.dtas has been saved.

◁

## Stored results

frames modify stores the following in r():

Scalars

| | |
|---|---|
| r(complevel) | compression level |
| r(compsize) | size, in bytes, of compressed file |
| r(compratio) | compression ratio, defined as the ratio of compressed size to uncompressed size |

Macros

| | |
|---|---|
| r(fn) | pathname of modified frameset file |
| r(frames) | list of frames in the modified frameset |
| r(added) | list of frames added, if add() specified |
| r(replaced) | list of frames replaced |
| r(dropped) | list of frames dropped, if drop() specified |

## Also see

[D] **frames save** — Save a set of frames on disk

[D] **frames use** — Load a set of frames from disk

[D] **frames** — Data frames